
Appendix A: The 2003 Social Accounting Matrix 

The Input- Output Table 

 

 A.AGRI A.MINE A.MAN A.TRADE A.SERV INF-A.AGRI INF-A.MINE INF-A.MAN INF-A.TRADE INF-A.SERV 

C.AGRI 
19 1 4757  13 22  1192   

C.MINE 
27 1 3942  416 30  987   

C.MAN 
825 44 7691 73 17860 947  1927   

C.TRADE 
          

C.SERV 
779 285 6806 7381 18000 894  1705   

INF-C.AGRI           

INF-C.MINE           

INF.C.MAN           

INF-C.TRADE           

INF-C.SERV           

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008) 

The Institutional Part of the Activity Columns 

 A.AGRI A.MINE A.MAN A.TRADE A.SERV INF-A.AGRI INF-A.MINE INF-A.MAN INF-A.TRADE INF-A.SERV 

SK-LAB 
193 6 1033 1271 7202 222  259   

SSK-LAB 
402 9 1526 2036 9784 461  382   

USK-LAB 
5874 58 4970 2868 10392 6744  1245  4885 

CAP 
1985 380 8144 7405 17264 2281  2041   

ENTR           

R-HHDS           

U-HHDS           

TRD           

TRE           

TRM           

GOVT           

YTAX           

VATB-RSA           

VATB-RoSADC           

VATB-ROW           

VATD           

REB 
-61 -15 -1230  -1872      

STAX           

ATAX 
 -1   -189      

MTAX-RSA           

MTAX-RoSADC           

MTAX-ROW           

S-I           

RSA           

RoSADC           

ROW           

TOTALS 10043 768 37639 21034 78870 11601 0 9738 0 4885 

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008) 

The Make Matrix 

 C.AGRI C.MINE C.MAN C.TRADE C.SERV INF-C.AGRI INF-C.MINE INF-C.MAN INF-C.TRADE INF-C.SERV 

A.AGRI 
10043 0 0 0 0      

A.MINE 
0 768 0 0 0      

A.MAN 
0 0 37639 0 0      

A.TRADE 
0 0 0 21034 0      

A.SERV 
0 0 0 0 78870      

INF-A.AGRI           

INF-A.MINE           

INF.A.MAN           

INF-A.TRADE           

INF-A.SERV           

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008) 

 

 

 

 

 

 

 

 

 

 

 

 



Appendix A 

 239 

 

The Institutional Part of the Activity Rows 

 SK-LSB SSK-LAB USK-LAB CAP ENTER R-HHDS U-HHDS TRD TRE TRM GOVT YTAX VATB-RSA VATB-RoSADC VATB-ROW 

A.AGRI 
 

        
      

A.MINE 
 

        
      

A.MAN 
 

        
      

A.TRADE 
 

        
      

A.SERV 
 

        
      

INF-A.AGRI      
11601 

         

INF-A.MINE      
0 

         

INF.A.MAN      
9738 

         

INF-A.TRADE      
0 

         

INF-A.SERV      
4885 

         

 

 

 VATD REB STAX ATAX MTAX-RSA MTAX-RoSADC MTAX-ROW S-I RSA RoSADC ROW TOTALS 

A.AGRI 
 

        
  

10043 

A.MINE 
 

        
  

768 

A.MAN 
 

        
  

37639 

A.TRADE 
 

        
  

21034 

A.SERV 
 

        
  

78870 

INF-A.AGRI            
11601 

INF-A.MINE            
0 

INF.A.MAN            
9738 

INF-A.TRADE            
0 

INF-A.SERV            
4885 

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008) 

The Institutional Part of the Commodity Rows 

 SK-LSB SSK-LAB USK-LAB CAP ENTER R-HHDS U-HHDS TRD TRE TRM GOVT YTAX VATB-RSA VATB-RoSADC VATB-ROW 

C.AGRI 
    

 
2114 6152 

  
      

C.MINE 
    

 
154 77 

  
      

C.MAN 
    

 
10180 22383 

  
      

C.TRADE 
    

 
  15783 1173 4078      

C.SERV 
    

 
5172 19747 

  
 14745     

INF-C.AGRI                

INF-C.MINE                

INF-C.MAN                

INF-C.TRADE                

INF-C.SERV                

 

 

 VATD REB STAX ATAX MTAX-RSA MTAX-RoSADC MTAX-ROW S-I RSA RoSADC ROW TOTALS 

C.AGRI 
    

   
66 706 270 2777 18089 

C.MINE 
    

   
476 32 12 127 6281 

C.MAN 
    

   
14708 3214 1231 12652 93735 

C.TRADE 
    

   
 0 0 0 21034 

C.SERV 
    

   
11783 4168 1351 3986 96802 

INF-C.AGRI            0 

INF-C.MINE            0 

INF-C.MAN            0 

INF-C.TRADE            0 

INF-C.SERV            0 

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008) 

The Institutional Part of the Commodity Columns 

 C.AGRI C.MINE C.MAN C.TRADE C.SERV INF-C.AGRI INF-C.MINE INF-C.MAN INF-C.TRADE INF-C.SERV 

SK-LAB 
          

SSK-LAB 
          

USK-LAB 
          

CAP 
          

ENTR 
          

R-HHDS           

U-HHDS           

TRD 
4814 353 11189  -573 

     

TRE 
743 11 419   

     

TRM 
472 10 3596   

     

GOVT           

YTAX           

VATB-RSA 22 4 1110  65      

VATB-RoSADC 3  151  4      

VATB-ROW 59 16 2481  1374      

VATD 
 39 1409  2579 

     

REB           

STAX 
  2468   

     

ATAX           

MTAX-RSA 7  619        

MTAX-RoSADC 1  84        

MTAX-ROW 40 5 1381        

S-I           

RSA 
473 1274 7828  656 

     

RoSADC 
53 142 873  41 

     

ROW 
1359 3659 22488  13786 

     

TOTALS 18089 6281 93735 21034 96802 0 0 0 0 0 

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008) 

The Institutional Diagonal Matrix  

 SK-LSB SSK-LAB USK-LAB CAP ENTER R-HHDS U-HHDS TRD TRE TRM GOVT YTAX VATB-RSA VATB-RoSADC VATB-ROW 

SK-LAB 
               

SSK-LAB 
               

USK-LAB 
               

CAP 
               

ENTR 
0 0 0 32281       146     

R-HHDS 
272 1899 31533 7219 3382 

     
117 

    

U-HHDS 
9914 12701 5503 0 22512 

     
294 

    

TRD 
     

       
   

TRE 
     

       
   

TRM 
     

       
   

GOVT     
102 

      3129 1201 158 3930 

YTAX     
925 133 2071 

        

VATB-RSA                

VATB-RoSADC                

VATB-ROW                

VATD 
     

       
   

REB                

STAX 
     

       
   

ATAX                

MTAX-RSA                

MTAX-RoSADC                

MTAX-ROW                

S-I     
1673 445 1837 

   
-1518 

    

RSA 
    3833 

       
   

RoSADC 
     

       
   

ROW 
     

       
   

TOTALS 10186 14600 37036 39500 32427 44422 52267 15783 1173 4078 13784 3129 1201 158 3930 

(The Institutional Diagonal Matrix continues) 

 VATD REB STAX ATAX MTAX-RSA MTAX-RoSADC MTAX-ROW S-I RSA RoSADC ROW TOTALS 

SK-LAB 
           10186 

SSK-LAB 
           14600 

USK-LAB 
           37036 

CAP 
           39500 

ENTR 
           32427 

R-HHDS            44422 

U-HHDS         
1343 

  52267 

TRD 
  

         15783 

TRE 
  

         1173 

TRM 
  

         4078 

GOVT 4027 -3178 2468 -190 626 85 1426     13784 

YTAX            3129 

VATB-RSA            1201 

VATB-RoSADC            158 

VATB-ROW            3930 

VATD 
  

         4027 

REB            -3178 

STAX 
  

         2468 

ATAX            -190 

MTAX-RSA            626 

MTAX-RoSADC            85 

MTAX-ROW            1426 

S-I         
4601 -1755 21750 

27033 

RSA 
  

         14064 

RoSADC 
  

         1109 

ROW 
  

         41292 

TOTALS 4027 -3178 2468 -190 626 85 1426 27033 14064 1109 41292  

Note: features are in Billion MT 

Source: Author’s own calculations based on Thurlow (2008). 

Appendix B: The IFPRI CGE Model Specifications 

 

Indices 

Index Variable definition 

s Formal activities 

    Aliases of s: g 

    Elements of s: 

           agri, 

           min, 

           man, 

           trade, 

           serv; 

is Informal activities 

    Aliases of is: ig 

    Elements of is: 

           agri, 

           min, 

           man, 

           trade, 

           serv; 

f Factors of production 

     Elements of f: 

           skl-lab, 

           sskl-lab, 

           uskl-lab, 

           cap; 

h Households 

       Elements of h: 

            r-hhds, 

            u-hhds; 

r Regions 

       Elements of r: 

            rsa, 

            rosadc, 

            row; 

Parameters 

Parameter Symbol Definition  

va (s)  Elasticity of substitution among primary factors in the production function 

of sector s 

sigmaT (s) σT
s Elasticity of transformation between domestic uses and exports in sector s 

relacet (s)  Elasticity of transformation among exports to different destinations 

sigmaQ (s) σQ
s Elasticity of substitution between domestic uses and imports of sector s 

relasarm (s)  Elasticity of substitution among imports from different origins 

at0 (s)  Activity tax rate for sector s  

rebt0 (s)  VAT rebate tax rate for sector s 

st0 (g)  Sales tax rate on commodity g 

vtd0 (g)  VAT collected domestically tax rate on commodity g  

th0 (h)  Household h income tax rate 

te0  Enterprise income tax rate 

mrge0 (g, r)  Share of export margins by commodity g and country of destination r 

tm0 (g, r)  Import tax rate on good g from region r 

vtb0 (g, r)  VAT collected at borders tax rate on good g imported from region r 

 

 

Prices 

Variable  Definition  

PY (g) Price index for domestic formal production s 

PIY (is) Domestic price for home- consumed commodity good g 

PD (g) Domestic price for marketed commodity good g 

PX (r, g) Producer price index for exports of commodity g to region r 

PA (g) Price index for aggregate Armington supply for good g 

PC (h) Consumption price for composite good g consumed by household h 

PT Price for transportation and marketing margins 

PG Price index for government expenditures 

PTRAN Price for social transfers 

PE Price for distributed profits 

PINV Investment price for composite good g 

PFX (r) Exchange rate respect region r 

PY0(s) Reference price for formal sector s production gross of activity tax 

PA0(g,s) Reference price for intermediate composite good g employed in sector s gross 

of VAT rebate 

Production 

Variable  Definition  

X(s) Domestic production of exports in sector s 

Y(s) Domestic total production (exports + domestic 

uses) in sector s 

IY(s) Home- consumption of commodity of sector s 

ES(r) Exports according to destination r 

A(s) Domestic output of composite activity s 

MG Domestic output of composite margins 

C(h) Household h total consumption 

G Government total consumption 

INVEST Domestic output of composite investment good  

 

 

Factors 

Variable Definition 

PF (f) Price for factor f 

FA0 (f, s) Factor f demand in activity s 

FIA0 (f, is) Factor f demand in activity is 

HF0 (h) Factor f income for household h 

EF0  Factor demand f income for enterprises 

 

 

Income and expenditures 

Variable  Definition  

CA0 (g,s) Intermediate demand of composite good g in 

sector s 

CIA0 (g, is) Intermediate demand of composite good g in 

sector is 

CH0 (g,h) Final demand of household h for marketed good g 

IAH0 (s, h) Final demand of household h for home consumed 

products of sector is 

C0 (h) Aggregate household h consumption 

HR0 (r,h) Remittances of household h from region r 

HSV0 (h,) Household h savings 

HE0 (h) Household h distributed profits 

HG0 (h) Household h social transfers 

TP0 (“YTAX”, h) Household h income tax 

Variable  Definition  

XE0 (r) Enterprise payments to region r 

ESV0 Enterprise savings 

EG0 Enterprise social transfers  

TE0 (“YTAX”, h) Enterprise income tax 

GD0 (g) Final government demand for composite good g 

FSV0 (r) Foreign savings from region r 

GSV0 Government savings 

GE0 Government distributed profits 

MTX0 (r, g) Import duties on good g from region r 

VTX0 (r, g) VAT on imports of good g from region r 

CS (g) Final investment demand for good g 

TA0 (“ATAX”, s) Activity tax in sector s  

TA0 (“REB”, s) VAT rebate in sector s 

TC0 (“VTD”, g) Domestic VAT on good g  

TC0 (“STAX”, g) Sales tax on good g 

 

 

Scaling parameters 

Parameter Definition  

fsav (r) Scaling parameter on foreign savings from region r 

gsav Scaling parameter on government savings  

Appendix C: The GAMS/MPSGE Code for the IFPRI Model  

In this appendix we present the GAMS/MPSGE code to perform and evaluate our simulations. 

This code is mainly compose of 3 parts: 

1) The SAM, presented in Appendix A, is called and imported into the GAMS code. Then, we 

check for SAM consistency imposing that rows sums equal columns sums (the notion of 

double- book keeping). Finally, we report if there are negative values or empty rows or 

columns. 

2) The SAM is manipulated and we move from a square matrix with numeric indices to model- 

relevant subtraces with meaningful text labels. This relies on the concept of “tuples”. 

According to Rutherford (2003) “these are multidimensional sets which can associated a 

numeric index with a text index”. In our case the SAM has 37 rows and columns  and in this 

part of the code it is broken down into 11 logical subsets, whose dimension is indicated in 

brackets: 

 

Production activities              A(5) 

Production informal activities     IA(5) 

Commodities                        C(5) 

Trade margins                      M(3) 

Primary factors                    F(4) 

Enterprises                        E(1) 

Private households                 H(2) 

Government                         G(1) 

Types of taxes                     T(7) 

Rest of world                      X(3) 

Investment- savings                I(1) 

 

This allows us to divide the SAM into sub- matrices as in the map below. Note that the same 

labels are employed later in the code. All cells with no labels are empty. 

 A IA C M F E H G T X I 

A   ac         

IA       iah     

C ca cia  cm   ch gd  er cs 

M   mc         

F fa           

E     ef   eg    

H     hf he  hg  hr  

G      ge   tr   

T ta tc    te tp     

X  rc    xe      

I      esv hsv gsv  fsv  

3)The model itself is constructed. Using the MPSGE solver, we need zero profit conditions 

(block $sectors:), market clearing conditions (block $commodities:), and income balance (block 

$consumers:). 

      

     

$TITLE An IFPRI CGE Model for Mozambique 

 

$ontext 

This model was developed to assess the impact of the Mozambican participation into the 

Southern Africa Development Community (SADC). This code is part of the Ph.D. thesis "The 

Mozambican Participation in SADC- A Liberalization Process trough Different Models and 

Different Closures". 

The study was conducted by Elisa Delpiazzo, under the supervision of Prof. Marco 

Missaglia, University of Pavia. The analysis uses a GAMS/MPSGE CGE model based on the 

2003 Mozambique SAM. The focus of the study is on examining the impact of a tariff 

reduction as the one scheduled in the SADC Trade Protocol. 

$offtext 

 

set colorder /chk/; 

 

set i SAM rows and column indices   /1*37/; 

 

*SAM rows and columns are the same set: 

 

alias (i,j); 

 

table sam(i,j)  Base year social accounts 

 

*Call the SAM from the worksheet in Excel format and import it in GAMS: 

 

$call=xls2gms  i=sam5sectors2hhds3reg.xls   o=sam5sectors2hhds3reg.inc 

$include sam5sectors2hhds3reg.inc 

; 

 

*Computation works best when features are around unit, so we scale the SAM values: 

 

sam(i,j)        = sam(i,j)/1000; 

 

display sam; 

 

*Check the consistency of the SAM and report negative values and empty rows or columns: 

 

parameter       samchk  Check of SAM consistency; 

samchk(i) = (sum(j, sam(i,j)-sam(j,i))); 

display samchk; 

set  negval(i,j)   flag for negative elements; 

negval(i,j) = yes$(sam(i,j)<0); 

display negval; 

 

set  empty(i,*)     flag for empty rows and columns; 

empty(i, "row")= 1$(sum(j,sam(i,j))=0); 

empty(j,"col")= 1$(sum(i,sam(i,j))=0); 

 

display empty; 

 

*Declare sets with which the SAM should be relabelled: 

 

set     s    List of formal activities and commodities/ 

                AGRI    agriculture fishing forestry and breeding activities, 

                MIN     mining, 

                MAN     industry (food processing and beverages light and heavy 

manufacturing metal products), 

                TRADE   wholesales and retail trade, 

                SERV    services/, 

 

        is   List of informal activities and commodities/ 

                INF-AGRI    agriculture fishing forestry and breeding activities, 

                INF-MIN     mining, 

                INF-MAN     industry (food processing and beverages light and heavy 

manufacturing metal products), 

                INF-TRADE   wholesales and retail trade, 

                INF-SERV    services/, 

 

        m    Margins/ 

                TMD     domestic sales transactions costs, 

                TMM     import transactions costs, 

                TME     export transactions costs/, 

 

        f    Factors of production/ 

                L1      unskilled labour, 

                L2      semi-skilled labour, 

                L3      skilled labour, 

                K       capital/, 

 

        h    Household / 

                R-HHDS  rural households, 

                U-HHDS  urban households /, 

 

        e    Enterprise /ENTR/, 

 

        t    Taxes/ 

                ATAX   Activity tax (subsidy), 

                REB    Vat rebate (on intermediate uses), 

                YTAX   Income tax (for households and private firms), 

                VATB   VAT tax collected at borders (on imports gross of tariffs), 

                VATD   VAT tax collected on domestic sales, 

                STAX   Sales tax, 

                MTAX   Import tax/, 

 

        r    Regions/ 

                RSA      Republic of South Africa, 

                ROSADC   Rest of SADC, 

                ROW      Rest of World/ 

 

alias  (i,j), (s,g), (is,ig), (m,mm); 

 

*Mapping the activities: 

 

Set     mapa(i,s) Mapping from SAM to formal activities/ 

                1.AGRI, 

                2.MIN, 

                3.MAN, 

                4.TRADE, 

                5.SERV /, 

 

        mapia(i,is) Mapping from SAM to informal activities/ 

                6.INF-AGRI, 

                7.INF-MIN, 

                8.INF-MAN, 

                9.INF-TRADE, 

                10.INF-SERV /, 

 

        mapc(i,g) Mapping from SAM to commodities/ 

                11.AGRI, 

                12.MIN, 

                13.MAN, 

                14.TRADE, 

                15.SERV /, 

 

        mapf(i,f) Mapping from SAM to factors of production / 

                16.L1, 

                17.L2, 

                18.L3, 

                19.K /, 

 

        mapm(i,m) Mapping from SAM to marketing margins / 

               20.TMD, 

               21.TMM, 

               22.TME /, 

 

        maph(i,h) Mapping from SAM to households / 

               23.R-HHDS, 

               24.U-HHDS /, 

 

        mape(i) Mapping from SAM to enterprise account /25/, 

 

        mapg(i)   Identifying government row           /26/, 

 

        mapt(i,t) Mapping from SAM to source of taxes / 

                 27.REB, 

                 28.ATAX, 

                 29.VATD, 

                 30.VATB, 

                 31.MTAX, 

                 32.STAX, 

                 33.YTAX /, 

 

         mapi(i)   Identifying the investment and saving row /34/, 

 

         mapx(i,r) Identifying the rest of the world / 

                 35.RSA, 

                 36.ROSADC, 

                 37.ROW /; 

 

set     ss/a, ia, c, m, f, e, h, g, t, x, i/ 

 

*Generate a report of submatrix totals: 

 

parameter     totals(*,*)        SAM totals for reporting; 

totals("a","c") = sum((mapa(i,s), mapc(j,g)), sam(i,j)); 

totals("ia","h") = sum((mapia(i,is), maph(j,h)), sam(i,j)); 

totals("c","a") = sum((mapc(i,g), mapa(j,s)), sam(i,j)); 

totals("c","ia") = sum((mapc(i,g), mapia(j,is)), sam(i,j)); 

totals("c","m") = sum((mapc(i,g), mapm(j,m)), sam(i,j)); 

totals("c","h") = sum((mapc(i,g), maph(j,h)), sam(i,j)); 

totals("c","g") = sum((mapc(i,g), mapg(j)), sam(i,j)); 

totals("c","x") = sum((mapc(i,g), mapx(j,r)), sam(i,j)); 

totals("c","i") = sum((mapc(i,g), mapi(j)), sam(i,j)); 

totals("m","c") = sum((mapm(i,m), mapc(j,g)), sam(i,j)); 

totals("f","a") = sum((mapf(i,f), mapa(j,s)), sam(i,j)); 

totals("f", "ia") = sum((mapf(i,f), mapia(j,is)), sam(i,j)); 

totals("e","f") = sum((mape(i),   mapf(j,f)), sam(i,j)); 

totals("e","g") = sum((mape(i),   mapg(j)), sam(i,j)); 

totals("h","f") = sum((maph(i,h), mapf(j,f)), sam(i,j)); 

totals("h","e") = sum((maph(i,h), mape(j)), sam(i,j)); 

totals("h","g") = sum((maph(i,h), mapg(j)), sam(i,j)); 

totals("h","x") = sum((maph(i,h), mapx(j,r)), sam(i,j)); 

totals("g","e") = sum((mapg(i),   mape(j)), sam(i,j)); 

totals("g","t") = sum((mapg(i),   mapt(j,t)), sam(i,j)); 

totals("t","a") = sum((mapt(i,t), mapa(j,s)), sam(i,j)); 

totals("t","c") = sum((mapt(i,t), mapc(j,g)), sam(i,j)); 

totals("t","e") = sum((mapt(i,t), mape(j)), sam(i,j)); 

totals("t","h") = sum((mapt(i,t), maph(j,h)), sam(i,j)); 

totals("x","c") = sum((mapx(i,r),   mapc(j,g)), sam(i,j)); 

totals("x","e") = sum((mapx(i,r),   mape(j)), sam(i,j)); 

totals("i","e") = sum((mapi(i),   mape(j)), sam(i,j)); 

totals("i","h") = sum((mapi(i),   maph(j,h)), sam(i,j)); 

totals("i","g") = sum((mapi(i),   mapg(j)), sam(i,j)); 

totals("i","x") = sum((mapi(i),   mapx(j,r)), sam(i,j)); 

 

alias (ss,sss); 

 

totals(ss,"total") = sum(sss, totals(ss,sss)); 

totals("total",ss) = sum(sss, totals(sss,ss)); 

 

option totals:1; 

display totals; 

 

*Extract sub- matrices from the SAM. When a sub- matrix is extracted the associated 

*values in the original SAM are set equal to zero. 

 

*Extraction of domestic production- related data from SAM: 

 

Parameter 

        ca0(g,s)    Intermediate inputs demand for formal sectors, 

        cia0(g,is)  Intermediate input demand for informal sectors, 

        fa0(f,s)    Factor demand (or value added) for formal sectors, 

        fia0(f,is)  Factor demand (or value added) for informal sectors, 

        ta0(t,s)    Tax collection, 

        iah0(is,h)  Household consumption of own production, 

        ac0(s,g)    Marketed output, 

        iac0(is)    production of informal sectors; 

 

loop((mapc(i,g), mapa(j,s)),     ca0(g,s) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapia(j,is)),   cia0(g,is) = sam(i,j); sam(i,j)=0;); 

loop((mapf(i,f), mapa(j,s)),     fa0(f,s) = sam(i,j); sam(i,j)=0;); 

loop((mapf(i,f), mapia(j,is)),   fia0(f,is) = sam(i,j); sam(i,j)=0;); 

loop((mapt(i,t), mapa(j,s)),     ta0(t,s) = sam(i,j); sam(i,j)=0;); 

loop((mapia(i,is), maph(j,h)),   iah0(is,h) = sam(i,j); sam(i,j)=0;); 

loop((mapa(i,s), mapc(j,g)),     ac0(s,g) = sam(i,j); sam(i,j)=0;); 

 

iac0(is) = sum(h,iah0(is,h)); 

*Extraction of commodity demand components from SAM: 

 

Parameter 

        mc0(m,g)        Marketing and transportation costs, 

        tc0(t,g)        Indirect taxes, 

        rc0(r,g)        Value of imports at cif price, 

        cm0(g,m)        Sales to wholesale and retail margins, 

        gd0(g)          Government demand, 

        er0(g,r)        Export (fob), 

        cs0(g)          Investment demand, 

        ch0(g,h)        Private consumption; 

 

loop((mapm(i,m), mapc(j,g)),    mc0(m,g) = sam(i,j); sam(i,j)=0;); 

loop((mapt(i,t), mapc(j,g)),    tc0(t,g) = sam(i,j); sam(i,j)=0;); 

loop((mapx(i,r), mapc(j,g)),    rc0(r,g) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapm(j,m)),    cm0(g,m) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapg(j)),      gd0(g) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapx(j,r)),    er0(g,r) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapi(j)),      cs0(g) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), maph(j,h)),    ch0(g,h) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of factor- related data from SAM: 

 

parameter 

        hf0(h,f)    Factors income to households, 

        ef0(f)      Factors income to enterprises; 

 

loop((maph(i,h), mapf(j,f)),      hf0(h,f) = sam(i,j); sam(i,j)=0;); 

loop((mape(i), mapf(j,f)),        ef0(f) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of household- related data from SAM: 

 

parameter 

        tp0(t,h)       Income tax payment by households 

        he0(h)         Enterprise payment to households 

        hg0(h)         Government transfer to households 

        hr0(h,r)       Household income from abroad 

        hsv0(h)        Household savings; 

 

loop((mapt(i,t), maph(j,h)),     tp0(t,h) = sam(i,j); sam(i,j)=0;); 

loop((maph(i,h), mape(j)),       he0(h) = sam(i,j); sam(i,j)=0;); 

loop((maph(i,h), mapg(j)),       hg0(h) = sam(i,j); sam(i,j)=0;); 

loop((maph(i,h), mapx(j,r)),     hr0(h,r) = sam(i,j); sam(i,j)=0;); 

loop((mapi(i), maph(j,h)),       hsv0(h) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of enterprise- related data from SAM: 

parameter 

        te0(t)          Enterprise direct taxes 

        ge0             Enterprise dividends paid to Government 

        xe0(r)          Enterprise payments to foreigners 

        esv0            Enterprise savings 

        eg0             Government transfers to enterprise; 

 

loop((mapt(i,t), mape(j)),      te0(t) = sam(i,j); sam(i,j)=0;); 

loop((mapg(i), mape(j)),        ge0 = sam(i,j); sam(i,j)=0;); 

loop((mapx(i,r), mape(j)),      xe0(r) = sam(i,j); sam(i,j)=0;); 

loop((mapi(i), mape(j)),        esv0 = sam(i,j); sam(i,j)=0;); 

loop((mape(i), mapg(j)),        eg0 = sam(i,j); sam(i,j)=0;); 

 

*Extraction of other saving- related data from SAM: 

 

parameter 

        gsv0       Government saving 

        fsv0(r)    Foreign saving; 

 

loop((mapi(i), mapg(j)),     gsv0 = sam(i,j); sam(i,j)=0;); 

loop((mapi(i), mapx(j,r)),   fsv0(r) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of other tax- related data from SAM: 

 

parameter 

        tr0(t)     Tax revenue; 

 

loop((mapg(i), mapt(j,t)),     tr0(t) = sam(i,j); sam(i,j)=0;); 

 

*Check if all values have been extracted from SAM: 

 

display "All values should be zero if all data has been read:", sam; 

 

display ca0,fa0,ta0,iah0,ac0,iac0, 

        mc0,tc0,rc0,cm0,gd0,er0,cs0,ch0, 

        hf0,ef0, 

        tp0,he0,hg0,hr0,hsv0, 

        te0,ge0,xe0,esv0,eg0, 

        gsv0,fsv0, 

        tr0; 

 

*Introduction of tax and trade- related data: 

 

table regtm(r,g) tariff revenue by country of origin and good 

                  AGRI         MIN      MAN 

RSA               0.007       0.001    0.618 

ROSADC            0.001           0    0.084 

ROW               0.040       0.004    1.382 

; 

 

table regvatb(r,g)   vat collected at borders by country of origin and good 

 

                   AGRI        MIN        MAN        SERV 

RSA               0.022       0.004      1.110       0.065 

ROSADC            0.003           0      0.151       0.004 

ROW               0.059       0.016      2.481       1.374 

; 

 

parameter 

    mtx0(r,g)    total import duties on imports of good g from region r, 

    tm0(g,r)     import tariff rate, 

    vtx0(r,g)    total vat collected at borders on good g from region r, 

    vtb0(g,r)    vat rate collected at borders on imports entering Armington supply; 

 

mtx0(r,g) = regtm(r,g); 

vtx0(r,g) = regvatb(r,g); 

tm0(g,r)$rc0(r,g) = mtx0(r,g)/rc0(r,g); 

vtb0(g,r)$rc0(r,g) = vtx0(r,g)/((1+tm0(g,r))*rc0(r,g)); 

 

display 

mtx0, vtx0, tm0, vtb0; 

 

*Manipulation of extracted values from SAM: 

 

parameter 

    at0(s)       activity tax (subsidies), 

    rebt0(s)     vat rebate (on intermediate uses), 

    st0(g)       sales tax, 

    vtd0(g)      vat collected on domestic sold commodities entering Armington supply, 

    th0(h)       household tax rate (on income from wages capital dividends and        

transfers), 

    et0          enterprise tax rate (on income from capital and transfers), 

    as0(g)       total Armington supply, 

    ex0(r,g)     exports of good G at producer price for region R, 

    dm0(g)       domestic uses, 

    ya0(g)       aggregate output from all sectors, 

    c0(h)        total household consumption (own + marketed), 

    ter0(g)      total exports of good G, 

    tex0(r)      total exports to region R, 

    mrge0(g,r)   export margin by country of destination and good 

; 

ter0(g) = sum(r, er0(g,r)); 

mrge0(g,r)$er0(g,r) = mc0("TME",g)*(er0(g,r)/ter0(g)); 

ex0(r,g) = er0(g,r) - mrge0(g,r); 

tex0(r) = sum(g, ex0(r,g)); 

ya0(g) = max(sum(s, ac0(s,g)), ter0(g)); 

dm0(g) = ya0(g) - (sum(r, ex0(r,g))); 

at0(s)$(not sum(g,ac0(s,g))=0) = ta0("ATAX",s)/sum(g,ac0(s,g)); 

rebt0(s)$(not(sum(g,ca0(g,s))=0)) = ta0("REB",s)/sum(g,ca0(g,s)); 

vtd0(g)$dm0(g) = tc0("VATD",g)/dm0(g); 

th0(h) = tp0("YTAX",h)/(sum(f,hf0(h,f)) + he0(h) + hg0(h) + (sum(r,hr0(h,r)))); 

et0 = te0("YTAX")/(sum(f,ef0(f)) + eg0); 

as0(g) = sum(r,rc0(r,g)) + sum(t,tc0(t,g)) + mc0("TMD",g) + mc0("TMM",g) + dm0(g); 

st0(g)$as0(g) = tc0("STAX",g)/as0(g); 

c0(h) = sum(g, ch0(g,h)) + sum(is, iah0(is,h)); 

 

display at0, rebt0, vtd0, th0, et0, as0, st0, ex0, ya0, dm0, ter0, tex0, mrge0; 

 

parameter 

py0(s)         Reference price of final activity products gross of activity tax, 

pa0(g,s)       reference price of intermediate uses gross of VAT rebate; 

 

py0(s)          = 1-at0(s); 

pa0(g,s)        = 1+rebt0(s); 

 

*Introduction of trade elasticities: 

 

table elasticity(*,*) elasticities in CES and CET functions 

 

                   sigmaQ     sigmaT      RELASARM    RELACET 

AGRI                2.1         2.1          5.8        5.8 

MIN                 3.1         3.1         13.2       13.2 

MAN                 2.6         2.6          6.7        6.7 

TRADE               1.9         1.9          3.8        3.8 

SERV                2.1         2.1          3.9        3.9 

; 

 

parameter 

 

sigmaQ(s)    Elasticity of substitution between imports and domestic uses, 

sigmaT(s)    Elasticity of transformation between exports and domestic uses, 

relasarm(s)  Elasticity of substitution among imports from different origins, 

relacet(s)   Elasticity of transformation among exports to different destinations; 

 

sigmaQ(s) = elasticity(s,"sigmaQ"); 

sigmaT(s) = elasticity(s,"sigmaT"); 

relasarm(s) = elasticity(s,"relasarm"); 

relacet(s) = elasticity(s, "relacet"); 

 

 

$ontext 

$model:moz 

 

$sectors: 

x(s)                          !allocation of output to domestic and export markets 

y(s)                          !sectoral output (domestic production) for formal sectors 

iy(is)$iac0(is)               !sectoral output (domestic production) for informal sectors 

es(r)$tex0(r)                 !export activity (applies margins) per destination 

a(s)                          !aggregate supply (armington aggregate) 

mg                            !transport margins 

c(h)                          !household consumption 

gd                            !government consumption 

invest                        !aggregate investment 

 

$commodities: 

py(g)                         !output price for formal sectors 

piy(is)$iac0(is)              !output price for informal sectors 

pd(g)$dm0(g)                  !domestic sales price 

px(r,g)$ex0(r,g)              !export price 

pa(g)                         !composite demand price for marketed output 

pc(h)                         !household consumption price 

pt                            !trade and transportation margins 

pf(f)                         !factor prices 

pg                            !government consumption price 

pe                            !enterprise rents 

pfx(r)                        !price of foreign exchange 

pinv                          !price of investments 

ptran                         !price of social transfers 

 

$consumers: 

ra(h)                         !private households 

entr                          !private firms 

govt                          !government 

foreign(r)                    !foreigners 

 

$auxiliary: 

fsav(r)      !scaling parameter for foreign savings from region r 

gsav         !scaling parameter for government savings 

 

*Sectoral production combines primary factors and intermediates. the first sector 

*produces market output and pays taxes, the second produces non- market output. 

 

$prod:y(s)      s:0  va:0.5 

o:py(g)         q:ac0(s,g)             a:govt    t:(at0(s))    p:py0(s) 

i:pa(g)         q:ca0(g,s)             a:govt    t:(rebt0(s))  p:pa0(g,s) 

i:pf(f)         q:fa0(f,s)    va: 

 

$prod:iy(is)$iac0(is)      s:0  va:0.5 

o:piy(is)         q:iac0(is) 

i:pa(g)           q:cia0(g,is) 

i:pf(f)           q:fia0(f,is)    va: 

 

*Commodity supply to domestic and export markets is modelled as a constant 

*elasticity of transformation function: 

 

$prod:x(g)      td:relacet(g)  t(td):sigmaT(g) 

o:pd(g)         q:dm0(g) 

o:px(r,g)       q:ex0(r,g) 

i:py(g)         q:ya0(g) 

 

*Domestic production for export markets (at producer price) is decomposed into 

*exports for different destinations (at final prices applying margins): 

 

$prod:es(r)$tex0(r) 

o:pfx(r)        q:(sum(g,er0(g,r))) 

i:px(r,g)       q:ex0(r,g) 

i:pt            q:(sum(g,mrge0(g,r))) 

 

*Production of marketing and transportation services: 

 

$prod:mg  s:0 

o:pt            q:(sum((m,g),mc0(m,g))) 

i:pa(g)         q:(sum(m, cm0(g,m))) 

 

*Supply of Armington composite supply involves collection of import duties and 

*other indirect taxes, the application of distribution margins: 

 

$prod:a(g)      ddm:relasarm(g)  dm(ddm):sigmaQ(g)   d:0   m(dm):0 

o:pa(g)         q:as0(g)               a:govt   t:st0(g) 

i:pd(g)         q:dm0(g)          d:   a:govt   t:vtd0(g) 

i:pt            q:mc0("TMD",g)    d: 

i:pfx(r)        q:rc0(r,g)        m:   a:govt   t:(vtb0(g,r)+(tm0(g,r)*(1+vtb0(g,r)))) 

i:pt            q:mc0("TMM",g)    m: 

 

*Investment demand: 

 

$prod:invest 

o:pinv          q:(sum(g,cs0(g))) 

i:pa(g)         q:cs0(g) 

 

*Household demand for market and non- market goods and services: 

$prod:c(h) 

o:pc(h)         q:(sum(g,ch0(g,h)) + sum(is,iah0(is,h))) 

i:pa(g)         q:ch0(g,h) 

i:piy(is)       q:iah0(is,h) 

 

*Government demand: 

$prod:gd 

o:pg            q:(sum(g,gd0(g))) 

i:pa(g)         q:gd0(g) 

 

*Household income and expenditures: 

 

$demand:ra(h) 

d:pc(h)        q:c0(h) 

e:pf(“l”)      q:(hf0(h,”l”)*(1-th(h)))                !labour income net of tax 

e:pf(“k”)      q:(hf0(h,”k”) *(1-th(h)))               !capital income net of tax 

e:pfx(r)       q:(hr0(h,r) *(1-th(h)))                 !remittances net of tax 

e:pinv         q:(-hsv0(h))                            !private savings 

e:pe           q:(he0(h) *(1-th(h)))                   !distributed profits net of tax 

e:ptran        q:(hg0(h) *(1-th(h)))                   !social transfers net of tax 

 

*Enterprise income and expenditures: 

 

$demand:entr 

d:pe           q:(sum(h,he0(h)) + ge0)       

e:pf("k")      q:(ef0("k")*(1-et0))                          !capital income net of tax 

e:pfx(r)       q:(-xe0(r))                                   !payments to foreigners  

e:ptran        q:(eg0*(1-et0))                               !social transfers net of tax 

e:pinv         q:(-(betaz)*(sum(g,cs0(g))))                  !enterprise savings  

 

*Government income and expenditures: 

 

$demand:govt 

d:pg           q:(sum(g,gd0(g))) 

e:pinv         q:(-gsv0)                           r:gsav     !government savings 

e:ptran        q:(-(eg0+sum(h,hg0(h))))                       !social trasnfers  

e:pe           q:ge0                                          !distributed profits 

e:pf("k")      q:(sum(h,(hf0("k",h)*(th0(h)))))               !tax on household capital 

e:pf("l")      q:(sum(h,(hf0("L",h)*(th0(h)))))               !tax on household labour 

e:pfx(r)       q:(sum(h,(hr0(h,r)*(th0(h)))))                 !tax on household   

remittances 

e:pe           q:(sum(h,(he0(h)*(th0(h)))))                   !tax on household 

distributed profits 

e:ptran        q:(sum(h,hg0(h)*(th0(h))))                     !tax on household social 

transfers 

e:pf("k")      q:(ef0("k")*(et0))                             !tax on enterprise capital 

e:ptran        q:(eg0*(et0))                                  !tax on enterprise social 

transfers 

 

*Foreigners’ income and expenditures: 

 

$demand:foreign(r) 

d:pfx(r)       q:(sum(g,er0(g,r))) 

e:pinv         q:(-fsv0(r))                     r:fsav(r)     !foreign savings 

e:pfx(r)       q:(xe0(r))                                     !enterprise payments to 

foreigners 

e:pfx(r)       q:(-sum(h,hr0(h,r)))                           !household remittances 

e:pfx(r)       q:(sum(g,rc0(r,g)))                            !imports 

 

*Complementarity condition for the scaling parameter fsav: 

 

$constraint:fsav(r) 

 

 fsav(r)*fsv0(r)*pinv =e= pfx(r)*((sum(g,rc0(r,g)))-(sum(g,er0(g,r)))+ (xe0(r))-   

sum(h,hr0(h,r))) ; 

 

*Complementarity condition for the scaling parameter gsav: 

 

$constraint:gsav 

 

 govt/pg =e= (sum(g,gd0(g))); 

 

$report: 

v:formprod(s,g) o:py(g)       prod:y(s) 

v:inforprod(is) o:piy(is)     prod:iy(is) 

v:import(r,g)   i:pfx(r)      prod:a(g) 

v:export(r,g)   o:px(r,g)     prod:x(g) 

v:exportfp(r)   o:pfx(r)      prod:es(r) 

v:domuses(g)    o:pd(g)       prod:x(g) 

v:margins       o:pt          prod:mg 

v:intermed(g,s) i:pa(g)       prod:y(s) 

v:factdeminf(f,is)  i:pf(f)   prod:iy(is) 

v:factdem(f,s)  i:pf(f)       prod:y(s) 

v:armington(g)  o:pa(g)       prod:a(g) 

v:rinvest       o:pinv        prod:invest 

v:rcons(h)      o:pc(h)       prod:c(h) 

v:rgovt         o:pg          prod:gd 

v:renter        d:pe          demand:entr 

 

$offtext 

$sysinclude mpsgeset  moz 

 

*Initialization of the scaling parameters and choice of the appropriate closure rule. 

km.L=1; 

 

*If the closure rule is EXOGENOUS foreign savings put an asterix before these commands: 

fsav.L(r)=1; 

fsav.LO(r)= -inf; 

**If the closure rule is ENDOGENOUS foreign savings put an asterix before the command: 

fsav.FX(r)=1; 

*If the closure rule is EXOGENOUS governmemnt savings put an asterix before these 

commands: 

gsav.L=1; 

gsav.LO=-inf; 

*If the closure rule is ENDOGENOUS government savings put an asterix before the command: 

gsav.FX=1; 

 

$include moz.gen 

solve moz using mcp; 

 

*Counterfactual: trade liberalization in the SADC area: 

 

tm0(g,"rsa") = 0; 

tm0(g,"rosadc") =0; 

 

$include moz.gen 

solve moz using mcp; 

Appendix D: The “Bastard Keynesian” CGE Model Specifications 

 

Indices 

Index Variable definition 

s Formal activities 

    Aliases of s: g 

    Elements of s: 

           agri, 

           min, 

           man, 

           trade, 

           serv; 

is Informal activities 

    Aliases of is: ig 

    Elements of is: 

           agri, 

           min, 

           man, 

           trade, 

           serv; 

f Factors of production 

     Elements of f: 

           lab, 

           cap; 

h Households 

       Elements of h: 

            r-hhds, 

            u-hhds; 

r Regions 

       Elements of r: 

            rsa, 

            rosadc, 

            row; 

Parameters 

Parameter Symbol Definition  

va (s)  Elasticity of substitution among primary factors in the production function 

of sector s 

sigmaT (s) σT
s Elasticity of transformation between domestic uses and exports in sector s 

relacet (s)  Elasticity of transformation among exports to different destinations 

sigmaQ (s) σQ
s Elasticity of substitution between domestic uses and imports of sector s 

relasarm (s)  Elasticity of substitution among imports from different origins 

at0 (s)  Activity tax rate for sector s  

rebt0 (s)  VAT rebate tax rate for sector s 

st0 (g)  Sales tax rate on commodity g 

vtd0 (g)  VAT collected domestically tax rate on commodity g  

th0 (h)  Household h income tax rate 

te0  Enterprise income tax rate 

mrge0 (g, r)  Share of export margins by commodity g and country of destination r 

tm0 (g, r)  Import tax rate on good g from region r 

vtb0 (g, r)  VAT collected at borders tax rate on good g imported from region r 

alphaz(h)  Share of household h savings on total investments 

betaz   Share of enterprise savings on total investments 

 

 

Prices 

Variable  Definition  

PY (g) Price index for domestic formal production s 

PIY (is) Domestic price for home- consumed commodity good g 

PD (g) Domestic price for marketed commodity good g 

PX (r, g) Producer price index for exports of commodity g to region r 

PA (g) Price index for aggregate Armington supply for good g 

PC (h) Consumption price for composite good g consumed by household h 

PT Price for transportation and marketing margins 

PG Price index for government expenditures 

PTRAN Price for social transfers 

PE Price for distributed profits 

PINV Investment price for composite good g 

PFX (r) Exchange rate respect region r 

PY0(s) Reference price for formal sector s production gross of activity tax 

PA0(g,s) Reference price for intermediate composite good g employed in sector s gross 

of VAT rebate 

 



Appendix D 

 265 

Production 

Variable  Definition  

X(s) Domestic production of exports in sector s 

Y(s) Domestic total production (exports + domestic 

uses) in sector s 

IY(s) Home- consumption of commodity of sector s 

ES(r) Exports according to destination r 

A(s) Domestic output of composite activity s 

MG Domestic output of composite margins 

C(h) Household h total consumption 

G Government total consumption 

INVEST Domestic output of composite investment good  

 

 

Factors 

Variable Definition 

PF (f) Price for factor f 

FA0 (f, s) Factor f demand in activity s 

FIA0 (f, is) Factor f demand in activity is 

HF0 (h) Factor f income for household h 

EF0  Factor demand f income for enterprises 

 

 

Income and expenditures 

Variable  Definition  

CA0 (g,s) Intermediate demand of composite good g in 

sector s 

CIA0 (g, is) Intermediate demand of composite good g in 

sector is 

CH0 (g,h) Final demand of household h for marketed good g 

IAH0 (s, h) Final demand of household h for home consumed 

products of sector is 

C0 (h) Aggregate household h consumption 

HR0 (r,h) Remittances of household h from region r 

HSV0 (h,) Household h savings 

HE0 (h) Household h distributed profits 

HG0 (h) Household h social transfers 

TP0 (“YTAX”, h) Household h income tax 

Variable  Definition  

XE0 (r) Enterprise payments to region r 

ESV0 Enterprise savings 

EG0 Enterprise social transfers  

TE0 (“YTAX”, h) Enterprise income tax 

GD0 (g) Final government demand for composite good g 

FSV0 (r) Foreign savings from region r 

GSV0 Government savings 

GE0 Government distributed profits 

MTX0 (r, g) Import duties on good g from region r 

VTX0 (r, g) VAT on imports of good g from region r 

CS (g) Final investment demand for good g 

TA0 (“ATAX”, s) Activity tax in sector s  

TA0 (“REB”, s) VAT rebate in sector s 

TC0 (“VTD”, g) Domestic VAT on good g  

TC0 (“STAX”, g) Sales tax on good g 

 

 

Scaling parameters 

Parameter Definition  

km Scaling parameter on labour supply 

fsav (r) Scaling parameter on foreign savings from region r 

gsav Scaling parameter on government savings  

Appendix E: The GAMS/MPSGE Code for the “Bastard Keynesian” 

Model 

 

$TITLE A “Bastard Keynesian” CGE model for Mozambique 

 

$ontext 

This model was developed to assess the impact of the Mozambican participation into the 

Southern Africa Development Community (SADC). This code is part of the Ph.D. thesis "The 

Mozambican Participation in SADC- A Liberalization Process trough Different Models and 

Different Closures". 

The study was conducted by Elisa Delpiazzo, under the supervision of Prof. Marco 

Missaglia, University of Pavia. The analysis uses a GAMS/MPSGE CGE model based on the 

2003 Mozambique SAM. The focus of the study is on examining the impact of a tariff 

reduction as the one scheduled in the SADC Trade Protocol. 

$offtext 

 

set colorder /chk/; 

 

set i SAM rows and column indices   /1*35/; 

 

*SAM rows and columns are the same set: 

 

alias (i,j); 

 

table sam(i,j)  Base year social accounts 

 

*Call the SAM from the worksheet in Excel format and import it in GAMS: 

 

$call=xls2gms  i=bksam.xls   o=bksam.inc 

$include bksam.inc 

; 

 

*Computation works best when features are around unit, so we scale the SAM values: 

 

sam(i,j)        = sam(i,j)/1000; 

 

display sam; 

 

*Check the consistency of the SAM and report negative values and empty rows or columns: 

parameter       samchk  Check of SAM consistency; 

samchk(i) = (sum(j, sam(i,j)-sam(j,i))); 

display samchk; 

 

set  negval(i,j)   flag for negative elements; 

negval(i,j) = yes$(sam(i,j)<0); 

display negval; 

 

set  empty(i,*)     flag for empty rows and columns; 

empty(i, "row")= 1$(sum(j,sam(i,j))=0); 

empty(j,"col")= 1$(sum(i,sam(i,j))=0); 

 

display empty; 

 

*Declare sets  with which the SAM should be relabbeled: 

 

set     s    List of formal activities and commodities/ 

                AGRI    agriculture fishing forestry and breeding activitities, 

                MIN     mining, 

                MAN  industry (food processing and beverages light and heavy 

manufacturing metal products), 

                TRADE   wholesales and retail trade, 

                SERV    services/, 

 

        is   List of informal activities and commodities/ 

                INF-AGRI   agriculture fishing forestry and breeding activitities, 

                INF-MIN    mining, 

                INF-MAN  industry (food processing and beverages light and heavy 

manufacturing metal products), 

                INF-TRADE  wholesales and retail trade, 

                INF-SERV   services/, 

 

        m    Margins/ 

                TMD     domestic sales transactions costs, 

                TMM     import transactions costs, 

                TME     export transactions costs/, 

 

        f    Factors of production/ 

                L       labour, 

                K       capital/, 

 

        h    Household / 

                RHHDS   rural households 

                UHHDS   urban households/, 

 

        e    Enterprise /ENTR/, 

 

        t    Taxes/ 

                ATAX   Activity tax (subsidy), 

                REB    Vat rebate (on intermediate uses), 

                YTAX   Income tax (for households and private firms), 

                VATB   VAT tax collected at borders (on imports gross of tariffs), 

                VATD   VAT tax collected on domestic sales, 

                STAX   Sales tax, 

                MTAX   Import tax/, 

 

        r    Regions/ 

                RSA      Republic of South Africa, 

                ROSADC   Rest of SADC, 

                ROW      Rest of World/ 

 

alias  (i,j), (s,g), (is,ig), (m,mm); 

 

*Mapping the activities: 

 

Set     mapa(i,s) Mapping from SAM to formal activities/ 

                1.AGRI, 

                2.MIN, 

                3.MAN, 

                4.TRADE, 

                5.SERV /, 

 

        mapia(i,is) Mapping from SAM to informal activities/ 

                6.INF-AGRI, 

                7.INF-MIN, 

                8.INF-MAN, 

                9.INF-TRADE, 

                10.INF-SERV /, 

 

        mapc(i,g) Mapping from SAM to commodities/ 

                11.AGRI, 

                12.MIN, 

                13.MAN, 

                14.TRADE, 

                15.SERV /, 

 

        mapf(i,f) Mapping from SAM to factors of production / 

                16.L, 

                17.K /, 

 

        mapm(i,m) Mapping from SAM to marketing margins / 

               18.TMD, 

               19.TMM, 

               20.TME /, 

 

        maph(i,h) Mapping from SAM to households / 

               21.RHHDS, 

               22.UHHDS/, 

        mape(i) Mapping from SAM to enterprise account /23/, 

 

        mapg(i) Identifying government row           /24/, 

 

        mapt(i,t) Mapping from SAM to source of taxes / 

                 25.REB, 

                 26.ATAX, 

                 27.VATD, 

                 28.VATB, 

                 29.MTAX, 

                 30.STAX, 

                 31.YTAX /, 

 

         mapi(i) Identifying the investment and saving row /32/, 

 

         mapx(i,r) Mapping from SAM to the rest of the world / 

                 33.RSA, 

                 34.ROSADC, 

                 35.ROW /; 

 

set     ss/a, ia, c, m, f, e, h, g, t, x, i/ 

 

*Generate a report of submatrix totals: 

 

parameter     totals(*,*)        SAM totals for reporting; 

totals("a","c") = sum((mapa(i,s), mapc(j,g)), sam(i,j)); 

totals("ia","h") = sum((mapia(i,is), maph(j,h)), sam(i,j)); 

totals("c","a") = sum((mapc(i,g), mapa(j,s)), sam(i,j)); 

totals("c","ia") = sum((mapc(i,g), mapia(j,is)), sam(i,j)); 

totals("c","m") = sum((mapc(i,g), mapm(j,m)), sam(i,j)); 

totals("c","h") = sum((mapc(i,g), maph(j,h)), sam(i,j)); 

totals("c","g") = sum((mapc(i,g), mapg(j)), sam(i,j)); 

totals("c","x") = sum((mapc(i,g), mapx(j,r)), sam(i,j)); 

totals("c","i") = sum((mapc(i,g), mapi(j)), sam(i,j)); 

totals("m","c") = sum((mapm(i,m), mapc(j,g)), sam(i,j)); 

totals("f","a") = sum((mapf(i,f), mapa(j,s)), sam(i,j)); 

totals("f", "ia") = sum((mapf(i,f), mapia(j,is)), sam(i,j)); 

totals("e","f") = sum((mape(i),   mapf(j,f)), sam(i,j)); 

totals("e","g") = sum((mape(i),   mapg(j)), sam(i,j)); 

totals("h","f") = sum((maph(i,h), mapf(j,f)), sam(i,j)); 

totals("h","e") = sum((maph(i,h), mape(j)), sam(i,j)); 

totals("h","g") = sum((maph(i,h), mapg(j)), sam(i,j)); 

totals("h","x") = sum((maph(i,h), mapx(j,r)), sam(i,j)); 

totals("g","e") = sum((mapg(i),   mape(j)), sam(i,j)); 

totals("g","t") = sum((mapg(i),   mapt(j,t)), sam(i,j)); 

totals("t","a") = sum((mapt(i,t), mapa(j,s)), sam(i,j)); 

totals("t","c") = sum((mapt(i,t), mapc(j,g)), sam(i,j)); 

totals("t","e") = sum((mapt(i,t), mape(j)), sam(i,j)); 

totals("t","h") = sum((mapt(i,t), maph(j,h)), sam(i,j)); 

totals("x","c") = sum((mapx(i,r),   mapc(j,g)), sam(i,j)); 

totals("x","e") = sum((mapx(i,r),   mape(j)), sam(i,j)); 

totals("i","e") = sum((mapi(i),   mape(j)), sam(i,j)); 

totals("i","h") = sum((mapi(i),   maph(j,h)), sam(i,j)); 

totals("i","g") = sum((mapi(i),   mapg(j)), sam(i,j)); 

totals("i","x") = sum((mapi(i),   mapx(j,r)), sam(i,j)); 

 

alias (ss,sss); 

 

totals(ss,"total") = sum(sss, totals(ss,sss)); 

totals("total",ss) = sum(sss, totals(sss,ss)); 

 

option totals:1; 

display totals; 

 

*Extract sub- matrices from the SAM. When a sub- matrix is extracted the associated  

*values in the original SAM are set equal to zero. 

 

*Extraction of domestic production- related data from SAM: 

 

Parameter 

        ca0(g,s)    Intermediate inputs demand for formal sectors, 

        cia0(g,is)  Intermediate input demand for informal sectors, 

        fa0(f,s)    Factor demand (or value added) for formal sectors, 

        fia0(f,is)  Factor demand (or value added) for informal sectors, 

        ta0(t,s)    Tax collection, 

        iah0(is,h)  Household consumption of own production, 

        ac0(s,g)    Marketed output, 

        iac0(is)    production of informal sectors; 

 

loop((mapc(i,g), mapa(j,s)),     ca0(g,s) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapia(j,is)),   cia0(g,is) = sam(i,j); sam(i,j)=0;); 

loop((mapf(i,f), mapa(j,s)),     fa0(f,s) = sam(i,j); sam(i,j)=0;); 

loop((mapf(i,f), mapia(j,is)),   fia0(f,is) = sam(i,j); sam(i,j)=0;); 

loop((mapt(i,t), mapa(j,s)),     ta0(t,s) = sam(i,j); sam(i,j)=0;); 

loop((mapia(i,is), maph(j,h)),   iah0(is,h) = sam(i,j); sam(i,j)=0;); 

loop((mapa(i,s), mapc(j,g)),     ac0(s,g) = sam(i,j); sam(i,j)=0;); 

 

iac0(is) = sum(h,iah0(is,h)); 

 

*Extraction of commodity demand components from SAM: 

 

Parameter 

        mc0(m,g)        Marketing and transportation costs, 

        tc0(t,g)        Indirect taxes, 

        rc0(r,g)        Value of imports at cif price, 

        cm0(g,m)        Sales to wholesale and retail margins, 

        gd0(g)          Government demand, 

        er0(g,r)        Export (fob), 

        cs0(g)          Investment demand, 

        ch0(g,h)        Private consumption; 

 

loop((mapm(i,m), mapc(j,g)),    mc0(m,g) = sam(i,j); sam(i,j)=0;); 

loop((mapt(i,t), mapc(j,g)),    tc0(t,g) = sam(i,j); sam(i,j)=0;); 

loop((mapx(i,r), mapc(j,g)),    rc0(r,g) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapm(j,m)),    cm0(g,m) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapg(j)),      gd0(g) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapx(j,r)),    er0(g,r) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), mapi(j)),      cs0(g) = sam(i,j); sam(i,j)=0;); 

loop((mapc(i,g), maph(j,h)),    ch0(g,h) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of factor- related data from SAM: 

 

parameter 

        hf0(f,h)    Factors income to households, 

        ef0(f)      Factors income to enterprises; 

 

loop((maph(i,h), mapf(j,f)),      hf0(f,h) = sam(i,j); sam(i,j)=0;); 

loop((mape(i), mapf(j,f)),        ef0(f) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of household- related data from SAM: 

 

parameter 

        tp0(t,h)         Income tax payment by households 

        he0(h)           Enterprise payment to households 

        hg0(h)           Government transfer to households 

        hr0(h,r)         Household income from abroad 

        hsv0(h)          Household savings; 

 

loop((mapt(i,t), maph(j,h)),     tp0(t,h) = sam(i,j); sam(i,j)=0;); 

loop((maph(i,h), mape(j)),       he0(h) = sam(i,j); sam(i,j)=0;); 

loop((maph(i,h), mapg(j)),       hg0(h) = sam(i,j); sam(i,j)=0;); 

loop((maph(i,h), mapx(j,r)),     hr0(h,r) = sam(i,j); sam(i,j)=0;); 

loop((mapi(i), maph(j,h)),       hsv0(h) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of enterprise- related data from SAM: 

parameter 

        te0(t)          Enterprise direct taxes 

        ge0             Enterprise dividends paid to Government 

        xe0(r)          Enterprise payments to foreigners 

        esv0            Enterprise savings 

        eg0             Government transfers to enterprise; 

loop((mapt(i,t), mape(j)),      te0(t) = sam(i,j); sam(i,j)=0;); 

loop((mapg(i), mape(j)),        ge0 = sam(i,j); sam(i,j)=0;); 

loop((mapx(i,r), mape(j)),      xe0(r) = sam(i,j); sam(i,j)=0;); 

loop((mapi(i), mape(j)),        esv0 = sam(i,j); sam(i,j)=0;); 

loop((mape(i), mapg(j)),        eg0 = sam(i,j); sam(i,j)=0;); 

 

*Extraction of other saving- related data from SAM: 

 

parameter 

        gsv0       Government saving 

        fsv0(r)    Foreign saving; 

 

loop((mapi(i), mapg(j)),     gsv0 = sam(i,j); sam(i,j)=0;); 

loop((mapi(i), mapx(j,r)),   fsv0(r) = sam(i,j); sam(i,j)=0;); 

 

*Extraction of other tax- related data from SAM: 

 

parameter 

        tr0(t)     Tax revenue; 

 

loop((mapg(i), mapt(j,t)),     tr0(t) = sam(i,j); sam(i,j)=0;); 

 

*Check if all values have been extracted from SAM: 

 

display "All values should be zero if all data has been read:", sam; 

 

display ca0,fa0,ta0,iah0,ac0,iac0, 

        mc0,tc0,rc0,cm0,gd0,er0,cs0,ch0, 

        hf0,ef0, 

        tp0,he0,hg0,hr0,hsv0, 

        te0,ge0,xe0,esv0,eg0, 

        gsv0,fsv0, 

        tr0; 

 

*Introduction of tax- and trade- related data: 

 

table regtm(r,g) tariff revenue by country of origin and good 

 

                  AGRI         MIN      MAN 

RSA               0.007          0    0.619 

ROSADC            0.001          0    0.084 

ROW               0.040      0.005    1.381 

; 

 

table regvatb(r,g)   vat collected at borders by country of origin and good 

                   AGRI        MIN        MAN        SERV 

RSA               0.022       0.004      1.110       0.065 

ROSADC            0.003           0      0.151       0.004 

ROW               0.059       0.016      2.481       1.374 

; 

 

parameter 

    mtx0(r,g)    total import duties on imports of good g from region r, 

    tm0(g,r)     import tariff rate, 

    vtx0(r,g)    total vat collected at borders on good g from region r, 

    vtb0(g,r)    vat rate collected at borders on imports entering Armington supply; 

 

mtx0(r,g) = regtm(r,g); 

vtx0(r,g) = regvatb(r,g); 

tm0(g,r)$rc0(r,g) = mtx0(r,g)/rc0(r,g); 

vtb0(g,r)$rc0(r,g) = vtx0(r,g)/((1+tm0(g,r))*rc0(r,g)); 

 

display 

mtx0, vtx0, tm0, vtb0; 

 

*Manipulation of extracted vlues from SAM: 

 

parameter 

    at0(s)       activity tax (subsidies), 

    rebt0(s)     vat rebate (on intermediate uses), 

    st0(g)       sales tax, 

    vtd0(g)      vat collected on domestic saled commodities entering Armington supply, 

    th0(h)      household tax rate (on income from wages capital dividends and 

transfers), 

    et0          enterprise tax rate (on income from capital and transfers), 

    as0(g)       total Armington supply, 

    ex0(r,g)     exports of good G at producer price for region R, 

    dm0(g)       domestic uses, 

    ya0(g)       aggregate output from all sectors, 

    c0(h)        total household consumtpion (own + marketed), 

    ter0(g)      total exports of good G, 

    tex0(r)      total exports to region R, 

    mrge0(g,r)   export margin by contry of destination and good, 

    alphaz(h)    household h share on total investments, 

    betaz        enterprise share in total investments, 

    sr(h)        saving propensity for household h 

; 

 

ter0(g) = sum(r, er0(g,r)); 

mrge0(g,r)$er0(g,r) = mc0("TME",g)*(er0(g,r)/ter0(g)); 

ex0(r,g) = er0(g,r) - mrge0(g,r); 

tex0(r) = sum(g, ex0(r,g)); 

ya0(g) = max(sum(s, ac0(s,g)), ter0(g)); 

dm0(g) = ya0(g) - (sum(r, ex0(r,g))); 

at0(s)$(not sum(g,ac0(s,g))=0) = ta0("ATAX",s)/sum(g,ac0(s,g)); 

rebt0(s)$(not(sum(g,ca0(g,s))=0)) = ta0("REB",s)/sum(g,ca0(g,s)); 

vtd0(g)$dm0(g) = tc0("VATD",g)/dm0(g); 

th0(h) = tp0("YTAX",h)/(sum(f,hf0(f,h)) + he0(h) + hg0(h) + (sum(r,hr0(h,r)))); 

et0 = te0("YTAX")/(sum(f,ef0(f)) + eg0); 

as0(g) = sum(r,rc0(r,g)) + sum(t,tc0(t,g)) + mc0("TMD",g) + mc0("TMM",g) + dm0(g); 

st0(g)$as0(g) = tc0("STAX",g)/as0(g); 

c0(h) = sum(g, ch0(g,h)) + sum(is, iah0(is,h)); 

alphaz(h) = hsv0(h)/sum(g, cs0(g)); 

betaz = esv0/sum(g,cs0(g)); 

sr(h) = hsv0(h)/((1-th0(h))*(sum(f,hf0(f,h)) + he0(h) + hg0(h) + (sum(r,hr0(h,r))))); 

 

display at0, rebt0, vtd0, th0, et0, as0, st0, ex0, ya0, dm0, ter0, tex0, mrge0, c0, 

alphaz, betaz, sr; 

 

parameter 

py0(s)        Reference price for final activity products gross of activity tax, 

pa0(g,s)      Reference price of intermediate uses gross of VAT rebate; 

 

py0(s)          = 1-at0(s); 

pa0(g,s)        = 1+rebt0(s); 

 

*Introduction of trade elasticities: 

 

table elasticity(*,*) elasticities in CES and CET functions 

 

                   sigmaQ     sigmaT      RELASARM    RELACET 

AGRI                2.1         2.1          5.8        5.8 

MIN                 3.1         3.1         13.2       13.2 

MAN                 2.6         2.6          6.7        6.7 

TRADE               1.9         1.9          3.8        3.8 

SERV                2.1         2.1          3.9        3.9 

; 

 

parameter 

 

sigmaQ(s)    Elasticity of substitution between imports and domestic uses, 

sigmaT(s)    Elasticity of transformation between exports and domestic uses, 

relasarm(s)  Elasticity of substitution among imports from different origins, 

relacet(s)   Elasticity of transformation among exports to different destinations 

; 

 

sigmaQ(s) = elasticity(s,"sigmaQ"); 

sigmaT(s) = elasticity(s,"sigmaT"); 

relasarm(s) = elasticity(s,"relasarm"); 

relacet(s) = elasticity(s, "relacet"); 

 

$ontext 

$model:moz 

 

$sectors: 

x(s)                          !allocation of output to domestic and export markets 

y(s)                          !sectoral output (domestic production) for formal sectors 

iy(is)$iac0(is)               !sectoral output (domestic production) for informal sectors 

es(r)$tex0(r)                 !export activity (applies margins) per destination 

a(s)                          !aggregate supply (armington aggregate) 

mg                            !transport margins 

c(h)                          !household consumption 

gd                            !government consumption 

invest                        !aggregate investments 

 

$commodities: 

py(g)              !output price for formal sectors 

piy(is)$iac0(is)   !output price for informal sectors 

pd(g)$dm0(g)       !domestic sales price 

px(r,g)$ex0(r,g)   !export price 

pa(g)              !composite demand price for marketed output 

pc(h)              !household consumption price 

pt                 !trade and transportation margins 

pf(f)              !factor prices 

pg                 !government consumption price 

pe                 !enterprise rents 

pfx(r)             !price of foreign exchange 

pinv               !price of investments 

ptran              !price of social transfers 

 

$consumers: 

ra(h)       !private households 

entr        !private firms 

govt        !government 

foreign(r)  !foreigners 

 

$auxiliary: 

Km           !Keynesian multiplier 

fsav(r)      !scaling parameter for foreign savings from region r 

gsav         !scaling parameter for government savings 

 

*Sectoral production combines primary factors and intermediates. The first sector  

*produces market output ad pays taxes, the second produces non- market output. 

 

$prod:y(s)      s:0  va:0.5 

o:py(g)         q:ac0(s,g)            a:govt    t:(at0(s))    p:py0(s) 

i:pa(g)         q:ca0(g,s)            a:govt    t:(rebt0(s))  p:pa0(g,s) 

i:pf(f)         q:fa0(f,s)    va: 

 

$prod:iy(is)$iac0(is)      s:0  va:0.5 

o:piy(is)         q:iac0(is) 

i:pa(g)           q:cia0(g,is) 

i:pf(f)           q:fia0(f,is)    va: 

 

*Commodity supply to domestic and export markets is modelled as a constant 

*elasticity of transformation function: 

 

$prod:x(g)      t:sigmaT(g)   tt(t):relacet(g) 

o:pd(g)         q:dm0(g) 

o:px(r,g)       q:ex0(r,g)    tt: 

i:py(g)         q:ya0(g) 

 

*Domestic production for export markets (at producer price) is decomposed 

*into exports for different destinations (at final prices applying margins): 

 

$prod:es(r)$tex0(r) 

o:pfx(r)        q:(sum(g,er0(g,r))) 

i:px(r,g)       q:ex0(r,g) 

i:pt            q:(sum(g,mrge0(g,r))) 

 

*Production of marketing and transportation services: 

 

$prod:mg  s:0 

o:pt            q:(sum((m,g),mc0(m,g))) 

i:pa(g)         q:(sum(m, cm0(g,m))) 

 

*Supply of Armington composite supply involves collection of import dulie and 

*other indirect taxes, the application of distribution margins: 

 

$prod:a(g)      ddm:relasarm(g)  dm(ddm):sigmaQ(g)   d:0   m(dm):0 

o:pa(g)    q:as0(g)                   a:govt       t:st0(g) 

i:pd(g)    q:dm0(g)         d:        a:govt       t:vtd0(g) 

i:pt       q:mc0("TMD",g)   d: 

i:pfx(r)   q:rc0(r,g)       m:       a:govt       t:(vtb0(g,r)+(tm0(g,r)*(1+vtb0(g,r)))) 

i:pt       q:mc0("TMM",g)   m: 

 

*Household demand for market and non- market goods and services: 

 

$prod:c(h) 

o:pc(h)         q:c0(h) 

i:pa(g)         q:ch0(g,h) 

i:piy(is)       q:iah0(is,h) 

*Government demand: 

 

$prod:gd 

o:pg            q:(sum(g,gd0(g))) 

i:pa(g)         q:gd0(g) 

 

*Investment demand: 

 

$prod:invest 

o:pinv          q:(sum(g,cs0(g))) 

i:pa(g)         q:cs0(g) 

 

*Household income and expenditures: 

 

$demand:ra(h) 

d:pc(h)        q:c0(h) 

e:pf("k")      q:(hf0("k",h)*(1-th0(h)))                    !capital income net of tax 

e:pf("l")      q:(hf0("L",h)*(1-th0(h)))      r:km          !labour income net of tax     

e:pfx(r)       q:(hr0(h,r)*(1-th0(h)))                      !remittances net of tax 

e:pe           q:(he0(h)*(1-th0(h)))                       !distributed profits net of 

tax  

e:ptran        q:(hg0(h)*(1-th0(h)))                        !social transfers net of tax 

e:pinv         q:(-alphaz(h)*(sum(g,cs0(g))))               !private savings 

 

*Enterprise income and expenditures: 

 

$demand:entr 

d:pe           q:(sum(h,he0(h)) + ge0)       

e:pf("k")      q:(ef0("k")*(1-et0))                          !capital income net of tax 

e:pfx(r)       q:(-xe0(r))                                   !payments to foreigners  

e:ptran        q:(eg0*(1-et0))                               !social transfers net of tax 

e:pinv         q:(-(betaz)*(sum(g,cs0(g))))                  !enterprise savings  

 

*Government income and expenditures: 

 

$demand:govt 

d:pg           q:(sum(g,gd0(g))) 

e:pinv         q:(-gsv0)                           r:gsav     !government savings 

e:ptran        q:(-(eg0+sum(h,hg0(h))))                       !social trasnfers  

e:pe           q:ge0                                          !distributed profits 

e:pf("k")      q:(sum(h,(hf0("k",h)*(th0(h)))))               !tax on household capital 

e:pf("l")      q:(sum(h,(hf0("L",h)*(th0(h)))))    r:km       !tax on household labour 

e:pfx(r)       q:(sum(h,(hr0(h,r)*(th0(h)))))                 !tax on household   

remittances 

e:pe           q:(sum(h,(he0(h)*(th0(h)))))                   !tax on household 

distributed profits 

e:ptran        q:(sum(h,hg0(h)*(th0(h))))                     !tax on household social 

transfers 

e:pf("k")      q:(ef0("k")*(et0))                             !tax on enterprise capital 

e:ptran        q:(eg0*(et0))                                  !tax on enterprise social 

transfers 

 

*Foreigners’ income and expenditures: 

 

$demand:foreign(r) 

d:pfx(r)       q:(sum(g,er0(g,r))) 

e:pinv         q:(-fsv0(r))                     r:fsav(r)     !foreign savings 

e:pfx(r)       q:(xe0(r))                                     !enterprise payments to 

foreigners 

e:pfx(r)       q:(-sum(h,hr0(h,r)))                           !household remittances 

e:pfx(r)       q:(sum(g,rc0(r,g)))                            !imports 

 

*Complementarity condition for the scaling parameter km: 

 

$constraint:km 

 

 sum(h,ra(h)) =e= sum(h,((1-sr(h))*(1-th0(h))*(hf0("l",h)*pf("l")+hf0("k",h)*pf("k") + 

he0(h) *pe+ hg0(h)*pg + (sum(r,hr0(h,r)*pfx(r)))))); 

 

*Complementarity condition for the scaling parameter fsav: 

 

$constraint:fsav(r) 

 

 fsav(r)*fsv0(r)*pinv =e= pfx(r)*((sum(g,rc0(r,g)))-(sum(g,er0(g,r)))+ (xe0(r))-   

sum(h,hr0(h,r))) ; 

 

*Complementarity condition for the scaling parameter gsav: 

 

$constraint:gsav 

 

 govt/pg =e= (sum(g,gd0(g))); 

 

$report: 

v:formprod(s,g) o:py(g)       prod:y(s) 

v:inforprod(is) o:piy(is)     prod:iy(is) 

v:import(r,g)   i:pfx(r)      prod:a(g) 

v:export(r,g)   o:px(r,g)     prod:x(g) 

v:exportfp(r)   o:pfx(r)      prod:es(r) 

v:domuses(g)    o:pd(g)       prod:x(g) 

v:margins       o:pt          prod:mg 

v:intermed(g,s) i:pa(g)       prod:y(s) 

v:factdeminf(f,is)  i:pf(f)   prod:iy(is) 

v:factdem(f,s)  i:pf(f)       prod:y(s) 

v:armington(g)  o:pa(g)       prod:a(g) 

v:rinvest       o:pinv        prod:invest 

v:rcons(h)      o:pc(h)       prod:c(h) 

v:rgovt         o:pg          prod:gd 

v:renter        d:pe          demand:entr 

 

$offtext 

$sysinclude mpsgeset  moz 

 

*Initialization of the scaling parameters and choice of the appropriate closure rule.  

 

km.L=1; 

 

*If the closure rule is EXOGENOUS foreign savings put an asterix before these commands: 

fsav.L(r)=1; 

fsav.LO(r)= -inf; 

**If the closure rule is ENDOGENOUS foreign savings put an asterix before the command: 

fsav.FX(r)=1; 

*If the closure rule is EXOGENOUS governmemnt savings put an asterix before these 

commands: 

gsav.L=1; 

gsav.LO=-inf; 

*If the closure rule is ENDOGENOUS government savings put an asterix before the command: 

gsav.FX=1; 

 

$include moz.gen 

solve moz using mcp; 

 

*Counterfactual: trade liberalization in the SADC area: 

 

tm0(g,"rsa") = 0; 

tm0(g,"rosadc") =0; 

 

$include moz.gen 

solve moz using mcp; 

Appendix F: The 2003 Structuralist/ Post- Keynesian SAM 

 

 Y IY Supply Marg Enter Rhhds Uhhds Govt Row Invest 

Y 148,354  -148,354        

IY  26,224    -26,224     

Supply -51,377 -5,743 171,805 -21,034  -9165 -25,154 -14,745 -30,526 -14,061 

w -53,185 -15,856    40,923 28,118    

r -29,617 -2,664   32,281      

Atax 190       -190   

VATreb 3,178       -3,178   

VATb -1,446 -162    -697 -1,914 5,289  -1,070 

Mtax -585 -65    -282 -773 2,137  -432 

VATd   -4,027     4,027   

Stax   -2,468     2,468   

Dtax     -925 -133 -2,071 3,129   

TMD   -15,783 15,783       

TMM -1,115 -125  4,078  -538 -1,475   -825 

TME   -1,173 1,173       

Row -14,397 -1,609   -3,833 -6,938 -19,043  56,465 -10,645 

SAV     -1,673 -445 -1,837 -1,518 -24,596 27033 

Transf     146 117 294 -577   

Rem       1,343  -1,343  

Dprof     25,996 3,382 22,512 102   

Source: Author’s own calculations 

Note: features are Billion MTs 

          The SAM above is expressed in its MCM format (see chapter 1 for an explanation of MCMs) 

Appendix G: The Structuralist/ Post- Keynesian CGE Model 

Specifications 

 

Indices 

Index Variable definition 

s Formal activities 

 Informal activities 

    Aliases of s: g 

h Households 

       Elements of h: 

            r-hhds, 

            u-hhds; 

 

 

Parameters 

Parameter Symbol Definition  

b(s)   Input- Output coefficient for labour in sector s 

shvaz (s)  Share of value added in sector s 

tau (s)  Mark- up rate in sector s 

a0 (s)  Input- Output coefficient for domestic intermediates used in sector s 

a1 (s)  Input- Output coefficient for imported intermediates (net of taxes and 

margins) in sector s 

at0 (s)  Activity tax rate for sector s  

rebt0 (s)  VAT rebate tax rate for sector s 

st0 (g)  Sales tax rate on commodity g 

vtd0 (g)  VAT collected domestically tax rate on commodity g  

th0 (h)  Household h income tax rate 

te0  Enterprise income tax rate 

mgm0   Share of margins per unitary imports 

tm0 (g, r)  Import tax rate on good g from region r 

vtb0 (g, r)  VAT collected at borders tax rate on good g imported from region r 

rho ρ Price elasticity in the final import demand function 

sigma  σ Income elasticty in the final import demand function 

beta1  Share of imports in total capital formation 

beta2  Share of domestic goods in total capital formation 

Prices 

Variable  Definition  

PY (g) Price index for domestic formal production s 

PA Final price for domestic products 

PC (h) Consumption price for composite good g consumed by household h 

PT Price for transportation and marketing margins 

PG Price index for government expenditures 

PTRAN Price for social transfers 

PE Price for distributed profits 

PINV Investment price for composite good g 

PFX (r) Exchange rate respect region r 

PIM Price of imports in domestic currency 

PY0(s) Reference price for formal sector s production gross of activity tax 

RPL(s) Reference price for labour employed in sector s 

RPI(s) Reference price for domestic intermediates 

RPII(s) Reference preice for imported intermediates 

Pstar Price of imports in foreign currency 

Pbar Price of exports in foreign currency 

 

 

Production 

Variable  Definition  

Y(s) Domestic total production in sector s 

FINIMP Final imports 

X Exports 

COMMX Domestic supply 

MG Domestic output of composite margins 

C(h) Household h total consumption 

GD Government total consumption 

INVEST Domestic output of composite investment good  

 

 

Factors 

Variable Definition 

W(s) Wage rate for labour employed in sector s 

FA0 (f, s) Factor f demand in activity s 

WAGEINC (h) Factor f income for household h 

Income and expenditures 

Variable  Definition  

DCH0 (h) Final demand of household h for domestic good  

CH0ME (s, h) Final demand of household h for home consumed 

products of sector is 

C0 (h) Aggregate household h consumption 

HR0 (h) Remittances of household h from abroad 

HSV0 (h,) Household h savings 

HE0 (h) Household h distributed profits 

HG0 (h) Household h social transfers 

XE0  Enterprise payments to foreigners 

ESV0 Enterprise savings 

EG0 Enterprise social transfers  

GD0 Final government demand  

DGD0 Final government demand for domestic goods 

FSV0 Foreign savings  

GSV0 Government savings 

GE0 Government distributed profits 

MTAX  Import duties on good g from region r 

VATB  VAT on imports of good g from region r 

DCS  Final investment demand for domestic good  

INV Aggregate total investments 

ATAX (s) Activity tax in sector s  

VATREB ( s) VAT rebate in sector s 

VATD Domestic VAT on good g  

STAX Sales tax on good g 

 

 

Foreign sector 

NMINT(s) Net Imported intermediate used in sector s 

MINT(s) Imported intermediates used in sector s (gross of taxes and margins) 

NMC0(h) Net household h consumption of imported goods  

MC0(h) Housheold h consumption of imported goods (gross of taxes and margins) 

NMCS Net imported goods in capital formation 

MCS Imported goods in capital formation (gross of taxes and margins) 

Scaling parameters 

Parameter Definition  

km Scaling parameter on labour supply (Keynesian multiplier) 

fsav Scaling parameter on foreign savings  

gsav Scaling parameter on government savings  

Appendix H: The GAMS/ MPSGE Code for the Structuralist/ Post- 

Keynesian Model 

$TITLE A Structuralist/ Post- Keynesian CGE Model for Mozambique 

 
$ontext 

This model was developed to assess the impact of the Mozambican participation into the 

Southern Africa Development Community (SADC). 

This code is part of the Ph.D. thesis "The Mozambican Participation in SADC- A 

Liberalization Process trough Different Models and Different Closures". 

The study was conducted by Elisa Delpiazzo, under the supervision of Prof. Marco 

Missaglia, University of Pavia. 

The analysis uses a GAMS/MPSGE CGE model based on the 2003 Mozambique SAM. 

The focus of the study is on examining the impact of a tariff reduction as the one 

scheduled in the SADC Trade Protocol. 
$offtext 

 

*Define the sets of the model, to use them in the vectorial language of the model: 

 

sets  s  /Y  formal production sector, 

          IY informal production sector/, 
 

      h  /RHHDS rural households, 

          UHHDS urban households/ 

alias(s,g) 

; 
 

*Data are inserted in a SAM (in MCM format: 

) 

table sam(*,*) 
 

              Y       IY     SUPPLY      MARG     ENTER    RHHDS     UHHDS     GOVT      ROW    INVEST 

Y          148.354          -148.354 

IY                  26.224                                -26.224 

SUPPLY     -51.377  -5.743   171.805    -21.034            -9.165   -25.154   -14.745  -30.526  -14.061 

W          -53.185 -15.856                                 40.923    28.118 

R          -29.617  -2.664                       32.281 

ATAX         0.190                                                             -0.190 

VATREB       3.178                                                             -3.178 

VATB        -1.446  -0.162                                -0.697   -1.914       5.289            -1.070 

MTAX        -0.585  -0.065                                -0.282   -0.773       2.137            -0.432 

VATD                          -4.027                                            4.027 

STAX                          -2.468                                            2.468 

DIRTAX                                           -0.925   -0.133   -2.071       3.129 

TMD                          -15.783     15.783 

TMM         -1.115  -0.125                4.078           -0.538   -1.475                        -0.825 

TME                           -1.173      1.173 

ROW        -14.397  -1.609                       -3.833   -6.938  -19.043               56.465  -10.645 

SAV                                              -1.673   -0.445   -1.837       1.518  -24.596 

TRANSF                                            0.146    0.117    0.294      -0.557 

REM                                                                 1.343               -1.343 

PROF                                            -25.996    3.382   22.512       0.102 

; 

 

*Declare parameters: 
 

parameters 

rho             Price elasticity in final import function /-0.418/ 

sigma           Income elasticity in final import function /1.553/ 

Pstar           foreign price of imports /1/ 

Pbar            foreign price of exports /1/ 

wz(s)           Sector s wage rate in the benchmark 

mx(s)           Marketed production of sector s 
nmx(s)          Non- marketed production of sector s 

gdpz(s)         Total production of sector s 

wbill(s)        Wage bill in sector s 
b(s)            Input- Output coefficient for labour of sector s 

shvaz(s)        Share of value added on GDP in sector s 

mrk(s)          Mark- up income in sector s 

tau(s)          mark- up rate in sector s 

mint(s)         Imported intermediates (gross of taxes and margins) used in sector s 

dint(s)         Domestic intermediate (gross of taxes and margins) used in sector s 

a0(s)           Domestic intermediate input- output coefficient in sector s 

a1(s)           Imported intermediate (net of taxes and margins) input- output   

coefficient in sector s 

atax(s)         Activity tax (subsidy) in sector s 

at0(s)          Activity tax rate in sector s 
vatreb(s)       VAT rebate on intermediates used in sector s 

rebt0(s)        VAT rebate rate on intermediates used in sector s 

nmint(s)        Net imported intermediates used in sector s 

vatb            Total VAT collected at borders 

vtb0            Tax rate of VAT collected at borders 

mtax            Total import duties 

tm0             Tax rate on imports 

fimp            Final imports (gross of taxes and margins) 

nfimp           Net final imports (cif prices) 

mrgm            Total margins on imports 
imrgm(s)        Total margins on intermediate imports (cif) used in sector s 

mgm0            Margins per unit of imports 

dm(s)           Ttal domestic uses 

exdp            Exports gross of margins 

vatd            Total VAT on domestic commodity 

vtd0            VAT rate on domestic commodities 

stax            Total sales tax 

st0             Sales tax rate 

mrgd            Total margins on domestic produced goods 
ts              Total supply of domestic final goods 
exp             exports net of margins 

mrge            Total margins on exports 

wageinc(h)      Wage income for housheold h 

hr0(h)          Remittances from abroad of household h 

mc0(h)          Consumption of imports of household h (gross) 

nmc0(h)         Net consumption of imports of household h 

c0(h)           Total consumption of household h 

dtax(h)         Total direct tax of household h 

th0(h)          Direct tax rate of household h 

hsv0(h)         Total savings of household h 
sh0(h)          Saving rate of household h 

hg0(h)          Total social transfers to household h 

he0(h)          Total distributed profits to household h 

dch0(h)         Total consumption of domestic good of household h 

chome(s,h)      Home consumption of goods of sector s of household h 

mkincome        Total mark- up income 

xe0             Total payments to foreigners 

dtaxe           Total direct tax for enterprises 

te0             Direct tax rate for enterprises 

esv0            Total savings enterprise 
se0             Saving rate for enterprise 

eg0             Total social transfers to enterprises 

gsv0            Government savings 

ge0             Total distributed profits to government 

dgd0            Total government expenditures in domestic goods 
gd0             Aggregate total government expenditures 

mcs             Total investments in imported goods (gross of taxes and margins) 

nmcs            Net total investments in imported goods 

inv             Total aggregate investments 

dcs             Total investments in domestic goods 

fsv0            Foreign savings 
py0(s)          Reference price for final production of sector s 

rpl(s)          Reference price for labour 

rpi(s)          Reference price intermediate domestic 

rpii(s)         Reference price intermediate imported 

beta1           Share of imports in total capital formation 

beta2           Share of domestic goods in total capital formation 

; 

 

*Extraction of data from the SAM. 

 
wz(s)=1; 

mx(s)= -sam(s,"supply"); 

nmx(s)= -sam(s,"rhhds"); 

wbill(s)= -sam("w",s); 

mrk(s)= -sam("r",s); 

nmint(s)= -sam("row",s); 

dint(s)= -sam("supply",s); 

atax(s)= sam("atax",s); 

vatreb(s)= sam("vatreb",s); 
vatb= sam("vatb","govt"); 

mtax= sam("mtax","govt"); 

imrgm(s) = -sam("tmm",s); 

mrgm= sam("tmm","marg"); 

dm(s)= -sam(s,"supply"); 

exdp= -sam("supply","row"); 

vatd= sam("vatd","govt"); 

stax= sam("stax","govt"); 

mrgd= sam("tmd","marg"); 

ts= sam("supply","supply"); 
mrge= sam("tme","marg"); 

wageinc(h)= sam("w",h); 

hr0(h)= sam("rem",h); 

nmc0(h)= -sam("row",h); 

dtax(h)= -sam("dirtax",h); 

hsv0(h)= -sam("sav",h); 

hg0(h)= sam("transf",h); 

he0(h)= sam("prof",h); 

dch0(h)= -sam("supply",h); 
chome(s,h)= -sam(s,h); 
mkincome= sam("r","enter"); 

xe0= -sam("row","enter"); 

dtaxe= -sam("dirtax","enter"); 

esv0= -sam("sav","enter"); 

eg0= sam("transf","enter"); 

gsv0= -sam("sav","govt"); 

ge0= sam("prof","govt"); 

dgd0= -sam("supply","govt"); 

nmcs= -sam("row","invest"); 

dcs= -sam("supply","invest"); 
fsv0= -sam("sav","row"); 

 

*Manipulation of data: 

 

gdpz(s)= mx(s)+nmx(s); 

mint(s)= nmint(s)- sam("mtax",s)- sam("vatb",s)- sam("tmm",s); 

nfimp= nmcs+(sum(h,nmc0(h))); 

fimp= nfimp-sum(h,sam("mtax",h))-sum(h,sam("vatb",h))-sum(h,sam("tmm",h))- 

           sam("mtax","invest")-sam("vatb","invest")-sam("tmm","invest"); 

mgm0= mrgm/(nfimp+sum(s,nmint(s))); 
exp= exdp-mrge; 

mc0(h)=nmc0(h)+(-sam("vatb",h))+(-sam("mtax",h))+(-sam("tmm",h)); 

mcs= nmcs+(-sam("mtax","invest"))+(-sam("vatb","invest"))+(-sam("tmm","invest")); 

c0(h)= dch0(h)+mc0(h)+(sum(s,chome(s,h))); 

inv= mcs+dcs; 
 

*Alternative definitions of tau: 

 

*tau(s) =((gdpz(s)+atax(s))/(gdpz(s)+atax(s)-mrk(s)))-1; 

tau(s)= mrk(s)/(wbill(s)+mint(s)+dint(s)-vatreb(s)); 

*tau(s)= (gdpz(s)+atax(s)-wbill(s)-mint(s)-dint(s)+vatreb(s))/(wbill(s)+mint(s)+dint(s)-
*vatreb(s)); 

 

*Manipulation of tax rates: 

 

at0(s)$mx(s)= atax(s)/mx(s); 

rebt0(s)= vatreb(s)/(mint(s)+dint(s)); 

tm0= mtax/(nfimp+(sum(s,nmint(s)))); 

vtd0$(not sum(s,dm(s))=0)= vatd/(sum(s,dm(s))); 

vtb0= vatb/(sum(s,nmint(s))+nfimp+mtax); 

st0= stax/ts; 
th0(h)= dtax(h)/(wageinc(h)+hr0(h)+hg0(h)+he0(h)); 

te0= dtaxe/(mkincome+eg0); 

 

*Definition of reference prices: 

py0(s) = 1-at0(s); 

rpl(s)= 1+tau(s); 

rpi(s)= 1-rebt0(s); 

rpii(s)= ((1+tm0)-rebt0(s)); 

 
*Parameters calibrated on the benchmark: 

 

b(s)= wbill(s)/(wz(s)*gdpz(s)); 

shvaz(s)= (wbill(s)+mrk(s))/gdpz(s); 

a0(s)= dint(s)/gdpz(s); 

a1(s)= nmint(s)/gdpz(s); 

sh0(h)= hsv0(h)/(wageinc(h)+hr0(h)+hg0(h)+he0(h)-dtax(h)); 

se0= esv0/(mkincome+eg0-dtaxe); 

beta1=mcs/inv; 

beta2=(1-beta1); 
 

*Show values for parameters: 

 

Display 

wz,mx,nmx,wbill,mrk,nmint,dint,atax,vatreb,vatb,mtax,imrgm,mrgm,dm, 

exdp,vatd,stax,mrgd,ts,mrge,wageinc,hr0,nmc0,dtax,hsv0,hg0,he0,dch0, 

chome,mkincome,xe0,dtaxe,esv0,eg0,gsv0,ge0,dgd0,nmcs,dcs,fsv0, 

gdpz,mint,nfimp,fimp,mgm0,exp,mc0,mcs,c0,inv,tau,at0,rebt0,tm0, 

vtd0,vtb0,st0,th0,te0,py0,rpl,rpi,rpii,b,shvaz,a0,a1,sh0,se0,beta1,beta2; 
 
$ontext 

$model:moz 

 

$sectors: 

y(s)           !domestic productive sectors (formal and informal) 

finimp         !production of final imports 

commx          !supply of domestic goods 

x              !production of exports at final prices 

mg             !production of wholesale and retail margins 

invest         !aggregate investment 
c(h)           !household consumption 

gd             !government consumption 

 

$commodities: 

py(s)          !output price for domestic production (formal and informal) 

w              !wage rate 

pfx            !exchange rate 

pim            !import price (in domestic currency) 

pt             !margin price 

pinv           !price of investments 
pg             !price of government consumption 

pe             !price of distributed profits 

pa             !price of marketed domestic production (gross of taxes and margins) 

pc(h)          !price of household consumption 

ptran          !price of social transfers 
 

$consumers: 

ra(h)          !households 

entre          !enterprises 

govt           !government 

foreign        !foreigners 
 

$auxiliary: 

km             !keynesian multiplier 

gsav           !scaling parameter for government savings 

fsav           !scaling parameter foreign savings 

FINALIMPORT    !final imports (to insert the demand function) 

INTIMP         !intermediate imports (to insert the demand function) 

 

 

*Sectoral production combines variable costs (labour, imported and domestic 
*intermediates) and a profit mark- up. 

*Formal sector earns activity subsidies as Activity tax and VAT rebate. 

$prod:y(s)      d:0  va:0 

o:py(s) q:gdpz(s)           p:py0(s)   a:govt    t:(-at0(s)) 

i:w     q:wbill(s)          p:rpl(s)   a:entre   t:((1-te0)*tau(s))  

                                        a:govt   t:(te0*tau(s)) 

i:pa    q:dint(s)           p:rpi(s)   a:entre   t:((1-te0)*tau(s)*(1-rebt0(s)))     
                                        a:govt   t:((te0*tau(s)*(1-rebt0(s)))-rebt0(s)) 

i:pfx   q:(nmint(s)*pstar)  p:rpii(s)   d:0      

                            a:entre  t:((1-te0)*tau(s)*((1+tm0)*(1+vtb0)+mgm0)*(1-rebt0(s))) 

a:govt   t:((te0*tau(s)*((1+tm0)*(1+vtb0)+mgm0)*(1-

rebt0(s)))+((1-rebt0(s))*(tm0*(1+vtb0)+vtb0)-rebt0(s)*(1+mgm0))) 

i:pt    q:imrgm(s)    d:0 

 

*Final imports is the aggregation of final uses of imports and transaction margins. 

*It involves the collection of a part of import duties and other indirect taxes: 

 
$prod:finimp 

o:pim    q:fimp 

i:pfx    q:(nfimp*pstar)            a:govt   t:(tm0*(1+vtb0)+vtb0) 

i:pt     q:(mrgm-(sum(s,imrgm(s)))) 

 

*Commodity supply of domestic products involves collection of indirect taxes and 

*the application of distribution margins: 

 

$prod:commx 
o:pa     q:ts                    a:govt t:st0 
i:py(s)  q:dm(s)                 a:govt t:vtd0 

i:pt     q:mrgd 

i:pt     q:mrge 

 

*Export transformation into foreign currency: 

 

$prod:x 

o:pfx    q:(Pbar*exdp) 

i:pa     q:exdp 

 
*Production of marketing and transportation margins: 

 

$prod:mg    s:0 

o:pt     q:(mrgd+mrge+mrgm) 

i:pa     q:(mrgd+mrge+mrgm) 

 

*Investment demand: 

 

$prod:invest  s:0 

o:pinv   q:inv 
i:pim    q:(beta1*inv) 

i:pa     q:(beta2*inv) 

 

*Household demand for non-marketed, domestic and imported goods and services: 

 
$prod:c(h) 

o:pc(h)  q:c0(h) 

i:pim    q:mc0(h) 

i:pa     q:dch0(h) 

i:py(s)  q:chome(s,h) 

 
*Government demand for goods and services: 

 

$prod:gd 

o:pg     q:gd0 

i:pa     q:dgd0 

 

*Household income and expenditures: 

 

$demand:ra(h) 

d:pc(h)  q:c0(h) 
e:w      q:((1-th0(h))*wageinc(h))       r:km    !net income out of wages 

e:ptran  q:((1-th0(h))*hg0(h))                   !net income out of social transfers 

e:pe     q:((1-th0(h))*he0(h))                   !net income out of distributed profits 

e:pfx    q:((1-th0(h))*hr0(h))                   !net income out of remittances 

e:pinv   q:(-hsv0(h))                           !private savings 

 

*Enterprises' income and expenditures: 

 

$demand:entre 
d:pe     q:(sum(h,he0(h))+ge0) 

e:pfx    q:(-xe0)                               !payments to foreigners 

e:pinv   q:(-esv0)                              !enterprises savings 

e:ptran  q:(eg0*(1-te0))                        !net income out of social transfers 

 

*Government income and expenditures: 

 

$demand:govt 

d:pg     q:gd0 

e:ptran  q:(-eg0-sum(h,hg0(h)))                       !social payments 
e:pe     q:ge0                                        !distributed profits 

e:pinv   q:(-gsv0)                         r:gsav     !governmemnt savings 

e:w      q:(sum(h,(th0(h)*wageinc(h))))    r:km       !direct taxes out of wage income of 

households 

e:ptran  q:(sum(h,th0(h)*hg0(h)))                     !direct taxes out of social 

transfers of households 

e:pe     q:(sum(h,(th0(h)*he0(h))))                   !direct taxes out of distributed 

profits of households 

e:pfx    q:(sum(h,(th0(h)*hr0(h))))                   !direct taxes out of remittances of 
households 

e:ptran  q:(te0*eg0)                                  !direct taxes out of social 

transfers of enterprises 

 

*Foreigners' income and expenditures: 

 

$demand:foreign 

d:pfx    q:(Pbar*exdp) 

e:pfx    q:xe0                                      !enterprise payments to foreigners 

e:pfx    q:(-sum(h,hr0(h)))                         !remittances 

e:pfx    q:(1)                  r:FINALIMPORT       !final imports 
e:pfx    q:(1)                  r:INTIMP            !intermediate imports 

e:pinv   q:(-fsv0)              r:fsav              !foreign savings 

 

*Complementarity condition for the scaling parameter km: 

 

$constraint:km 

sum(h,ra(h)) =e= sum(h,((1-sh0(h))*(1-

th0(h))*(w*km*wageinc(h)+ptran*hg0(h)+pe*he0(h)+pfx*hr0(h)))); 

 

*Complementarity condition for the scaling parameter gsav: 
 

$constraint:gsav 

govt/pg =e= gd0; 

 

*Complementarity condition for the scaling parameter fsav: 
 

$constraint:fsav 

 

fsav*fsavz*pinv =e= pfx*((nfimp*Pstar)+(sum(s,nmint(s)*Pstar))-(Pbar*exdp))+pfx*(xe0-

sum(h,hr0(h))); 

 
*Constraint to define the functional form of the intermediate import demand: 

 

$constraint:INTIMP 

INTIMP =e= sum(s,(a1(s)*Pstar*gdpz(s)*y(s))); 

 

*Constraint to define the functional form of the intermediate import demand: 

 

$constraint: FINALIMPORT 

FINALIMPORT =e= 

(nfimp/(sum(s,gdpz(s)*y(s)))**sigma)*(pfx*Pstar/pa)**rho*(sum(s,gdpz(s)*y(s)))**sigma; 
 

$report: 

v:rgdp(s)    o:py(s)    prod:y(s) 

v:rlab(s)    i:w        prod:y(s) 

v:realint(s) i:pfx      prod:y(s) 

v:rfinimp    i:pfx      prod:finimp 

v:rexp       o:pfx      prod:x 

v:totmrg     o:pt       prod:mg 

v:rinv       o:pinv     prod:invest 
v:rcons(h)   o:pc(h)    prod:c(h) 

v:rgovt      o:pg       prod:gd 

v:prof       d:pe       demand:entre 

 

$offtext 

$sysinclude mpsgeset moz 

 

*Initialization of the scaling parameters and choice of the appropriate closure rule. 

 

km.L = 1; 
 

*Inizialization of the value of INTIMP. 

 

INTIMP.L=sum(s,nmint(s)); 

 

*Initialization of the value of FINALIMPORT. 

 

FINALIMPORT.L=nfimp; 

 
*If the closure rule is EXOGENOUS foreign savings put an asterix before these commands: 
fsav.L = 1; 

fsav.LO = -inf; 

*If the closure rule is ENDOGENOUS foreign savings put an asterix before the command: 

fsav.FX = 1; 

*If the closure rule is EXOGENOUS governmemnt savings put an asterix before these 

commands: 

gsav.L = 1; 

gsav.LO = -inf; 

*If the closure rule is ENDOGENOUS government savings put an asterix before the command: 

gsav.FX = 1; 
 

*Numeraire of the model: 

w.FX = 1; 

 

$include moz.gen 

solve moz using mcp; 

 

*Counterfactual: Trade liberalization in the SADC area. 

 

mtx= 0.027; 
$include moz.gen 

Solve moz using MCP; 

 


